
Modellierung und Analyse von Stromrichterschaltungen 

Gleichrichter in Zweipuls-Mittelpunktschaltung 

  
Das Bild zeigt die Schaltung des Gleichrichters (M2-Schaltung) mit ohmsch-induktiver Last 
(R3, L3). Die Induktivitäten L1 und L2 berücksichtigen zusammenfassend die Streureaktanzen 
des Stromrichtertransformators, die Reaktanzen der Kommutierungsdrosseln und des 
speisenden Netzes. Die ohmschen Widerstände der Transformatorwicklungen, der Drosseln 
und der Leitungen repräsentieren die Widerstände R1 und R2. In Abhängigkeit vom 
Schaltzustand der beiden Dioden kann man drei Fälle unterscheiden: 

              Diode 1      Diode 2    
Fall 1:    leitend        sperrend 
Fall 2:    sperrend     leitend   
Fall 3:    leitend        leitend  

Fall 3 ist der Kommutierungszustand. Unter dem Einfluss von Netz- und Lastspannung geht 
eine der Dioden in den Zustand "leitend" über, während die andere noch Strom führt. Zwischen 
diesen Zweigen fließt daher ein Kommutierungsstrom, der den Strom in einem der Zweige ab- 
und im anderen Zweig aufbaut.  
Die Dioden werden im vorliegenden Beispiel als ideale Schalter betrachtet, d. h. sie führen im 
Sperrzustand keinen Strom. Beim Aufstellen des Modells für die genannten Fälle kann man 
daher den Zweig mit einer sperrenden Diode als nicht existent ansehen. Somit wird jeder Fall 
durch ein separates Gleichungssystem beschrieben. Nach dieser Vorgehensweise erhält man 
mit dem Lastzweig als vollständigen Baum [1, Kapitel 4]  folgende Maschengleichungen: 

Fall 1:   

Fall 2:   

Fall 3: 

       

      

 
Die Maschengleichungen M1 und M2 liefern durch Umstellung die Differentiale di1/dt für den 
Fall 1 und di2/dt für den Fall 2. Die beiden Maschengleichungen zum Fall 3 haben weitgehende 
Ähnlichkeit mit denen des  Beispiels "Drehstromdrossel" im Abschnitt 3.4.1 von [1]. In jeder 
Gleichung erscheinen die Ableitungen der beiden Ströme i1 und i2. Durch einfaches Umstellen 
der Gleichungen nach einer der Ableitungen ist keine Differentialgleichung zu gewinnen, für 
deren rechte Seite alle Anfangswerte bekannt sind (algebraische Schleife). Dieses Problem 
lässt sich jedoch wie in [1] durch Auflösung des linearen Gleichungssystems nach den beiden 
ersten Ableitungen von i1 und i2 beheben.  
Maple-Programm: 
> restart:  
> M1:= (L1 + L3)*diff(i1(t),t) + (R1 + R3)*i1(t) = u1(t); 

 

> M2:= (L2 + L3)*diff(i2(t),t) + (R2 + R3)*i2(t) = u2(t); 

 

> M3_1:= (L1+L3)*diff(i1(t),t)+(R1+ R3)*i1(t) +   
          R3*i2(t)+L3*diff(i2(t),t) = u1(t); 

 

> M3_2:= (L2+L3)*diff(i2(t),t)+(R1+ R3)*i2(t) + 
          R3*i1(t)+L3*diff(i1(t),t) = u2(t); 

 

Umstellung der Maschengleichungen M1 und M2: 
> G1:= isolate(M1, diff(i1(t),t)); 

 

> G2:= isolate(M2, diff(i2(t),t)); 

Auflösung der durch M3_1 und M3_2 gebildeten algebraischen Schleife: 
> G3:= solve({M3_1, M3_2}, [diff(i1(t),t), diff(i2(t),t)]); 

 

> collect(G3, [i1(t),i2(t)]);  

Die Spannung uL über dem Lastzweig wird beschrieben durch 
> G4:= uL(t) = R3*(i1(t)+i2(t)) + 
               L3*(diff(i1(t),t)+diff(i2(t),t)); 

 

Weil jede der Gleichungen nur unter bestimmten Bedingungen gültig ist, müssen noch 
geeignete Schaltbedingungen formuliert werden. Dazu werden die logischen Größen D1 und 
D2 eingeführt. D1 hat den Wert "wahr" (true), wenn Diode 1 leitend ist, also im Fall 1, sonst 
den Wert "falsch". Sinngemäß gilt das für D2 mit Diode 2. Leitend ist eine Diode, wenn sie 
Strom führt (iV  > 0) oder wenn nach dem Sperrzustand wieder die Bedingung für das Einsetzen 
des Stronmflusses erfüllt ist (u1 > uL  bzw. u2 > uL).  
Im Programm wird jedoch wie in [2] statt uL die geringfügig verzögerte Spannung uLv  zur 
Bildung von D1 und D2 verwendet, um algebraische Schleifen über i1, i2 und uL  zu vermeiden. 
uLv wird durch ein Verzögerungsglied 1. Ordnung gebildet. 
> DG1:= diff(uLv(t),t) = (uL(t)-uLv(t))/0.00001; 

 

> D1:= (i1(t)>0) or (u1(t) > uLv(t)); 
 

> D2:= (i2(t)>0) or (u2(t) > uLv(t)); 

Mit Hilfe des Befehls piecewise werden nun die Differentialgleichungen für di1/dt und di2/dt 
stückweise aus den oben für die einzelnen Fälle ermittelten Gleichungen zusammengesetzt. 
Dabei ist darauf zu achten, dass die Auswertung der Bedingungen von links nach rechts erfolgt 
und nach der ersten gültigen Bedingung beendet wird. 
> DG2:= diff(i1(t),t) = piecewise(D1 and D2, rhs(G3[1,1]),  
                                 D1 and (not(D2)), rhs(G1), 0): 
> DG3:= diff(i2(t),t) = piecewise(D1 and D2, rhs(G3[1,2]),  
                                 D2 and (not(D1)), rhs(G2), 0): 
Vor der Lösung des Gleichungssystems mit dsolve sind noch die Gleichungen für die Netz-
spannungen u1 und u2 zu notieren und die Werte der Parameter festzulegen. 
> G5:= u1(t) = sqrt(2)*Ueff*sin(omega*t); 

 
> G6:= u2(t) = -sqrt(2)*Ueff*sin(omega*t); 

 
> Ueff:=100: omega:=314: R1:=2: R2:=2: R3:=10: L1:=0.04: 
  L2:=0.04: L3:=0.2: 
> Loe:= dsolve({DG1,DG2,DG3,G4,G5,G6,i1(0)=0,i2(0)=0,uLv(0)=0}, 
              numeric, [uL(t), i1(t),i2(t),u1(t),u2(t),uLv(t)], 
              output=listprocedure, interr=false, maxfun=0);  

Die im Befehl dsolve verwendete Option interr = false wird im Abschnitt 3.4.5 von [1]  be-
schrieben. Ohne diese Angabe kommt es zu einem vorzeitigen Abbruch der numerischen 
Rechnung, weil die Zwischenergebnisse eine Singularität vortäuschen. 
Die Teillösungen werden nun Variablen zugewiesen und anschließend graphisch dargestellt. 
> i1a:= subs(Loe, i1(t)): 
> i2a:= subs(Loe, i2(t)): 
> uLa:= subs(Loe, uL(t)): 
> plot([i1a,i2a], 0..0.06, labels=["t/s", "i1/A, i2/A"], 
        legend=["i1(t)", "i2(t)"]); 

> with(plots): 
> p1:= plot(i1a+i2a, 0..0.06, color=blue, labels=["t/s", 
            "iL/A"], legend=["iL(t)"]): 
> p2:= plot(uLa, 0..0.06, color=black, labels=["t/s", "uL/V"], 
            legend=["uL(t)"]): 
> dualaxisplot(p1,p2); 

